Passive Second Order Low Pass Filter

Sam Protsenko

February 9, 2017

Abstract

Article shows calculating of passive 2nd order LPF (Low Pass Filter) parameters, using Linux software like `Maxima` and `Qucs`.

1 Transfer function calculation

1.1 Calculating output voltage



Figure 1: Passive 2nd order LPF

Let's find out v_O using **node method**. All currents:

$$i_1 = \frac{v_I - v_{C_1}}{R_1}$$

$$i_2 = \frac{v_{C_1}}{X_{C_1}} = \frac{v_{C_1}}{\frac{1}{sC_1}}$$

$$i_3 = \frac{v_{C_1}}{R_2 + X_{C_2}} = \frac{v_{C_1}}{R_2 + \frac{1}{sC_2}}$$

KCL for v_{C_1} node:

$$i_1 - i_2 - i_3 = 0$$

Substitute currents equations to KCL and solve it for v_{C_1}:

$$v_{C_1} = v_I \frac{R_2 C_2 s + 1}{(R_1 C_1 s + 1)(R_2 C_2 s + 1) + R_1 C_2 s}$$

Output voltage:

$$v_O = v_{C_1} - i_3 R_2 = v_I \frac{1}{(R_1 C_1 s + 1)(R_2 C_2 s + 1) + R_1 C_2 s}$$

Transfer function:

$$\begin{aligned} H(s) = \frac{v_O}{v_I} &= \frac{1}{(R_1 C_1 s + 1)(R_2 C_2 s + 1) + R_1 C_2 s} \\ &= \frac{1}{R_1 R_2 C_1 C_2 s^2 + (R_1 C_1 + R_1 C_2 + R_2 C_2) s + 1} \\ &= \frac{1}{m s^2 + b s + k} \end{aligned}$$

Where coefficients:

$$\begin{aligned} m &= R_1 R_2 C_1 C_2 \\ b &= R_1 C_1 + R_1 C_2 + R_2 C_2 \\ k &= 1 \end{aligned}$$

Transfer function was calculation in `Maxima`, see listing 1.

Listing 1: Transfer function calculation

```
i1: (vI - vC1) / R1$
i2: vC1 / (1/s/C1)$
i3: vC1 / (R2 + 1/s/C2)$
eq1: i1 - i2 - i3 = 0$
sol: solve(eq1, vC1)$
vC1: rhs(sol[1]);
i1: ev(i1, sol)$
i2: ev(i2, sol)$
i3: ev(i3, sol)$
vO: ratsimp(vC1 - i3*R2);
define(H(s), vO/vI);
```

1.2 Standard form of transfer function

Standard form of transfer function for second order LPF:

$$H(s) = \frac{k_{dc}}{\frac{s^2}{\omega_n^2} + 2\frac{\zeta}{\omega_n}s + 1} = \frac{k_{dc}\omega_n^2}{s^2 + 2\zeta\omega_n s + \omega_n^2}$$

Let's calculate coefficients for standard form from our transfer function:

$$\begin{aligned}
k_{dc} &= \frac{1}{k} && = 1 \\
\omega_n &= \sqrt{\frac{k}{m}} && = \frac{1}{\sqrt{R_1 R_2 C_1 C_2}} \\
\zeta &= \frac{b}{2m\sqrt{k/m}} && = \frac{R_1 C_1 + R_1 C_2 + R_2 C_2}{2\sqrt{R_1 R_2 C_1 C_2}}
\end{aligned}$$

where

- k_{dc} – gain
- ω_n – natural frequency
- ζ – damping ratio

Also, quality factor is:

$$Q = \frac{1}{2\zeta} = \frac{m\sqrt{k/m}}{b} = \frac{\sqrt{R_1 R_2 C_1 C_2}}{R_1 C_1 + R_1 C_2 + R_2 C_2}$$

Let's check if calculations are correct so far (see listing 2).

Listing 2: Verify if transfer function in standard form is correct

```
omega_n: 1 / sqrt(R1*R2*C1*C2)$
zeta: (R1*C1+R1*C2+R2*C2) / (2*sqrt(R1*R2*C1*C2))$
H(s) := omega_n^2 / (s^2 + 2*zeta*omega_n*s + omega_n^2)$
expand(H(s));
```

2 Cut-off frequency calculation

From Laplace transform to Fourier transform:

$$H(j\omega) = H(s)\Big|_{s=j\omega} = \frac{\omega_n^2}{-\omega^2 + 2\zeta\omega_n j\omega + \omega_n^2}$$

Real part:

$$\mathrm{Re}(H(j\omega)) = \frac{\omega_n^2(\omega_n^2 - \omega^2)}{(\omega_n^2 - \omega^2)^2 + 4\zeta^2\omega^2\omega_n^2}$$

Imaginary part:

$$\mathrm{Im}(H(j\omega)) = -\frac{2\zeta\omega\omega_n^3}{(\omega_n^2 - \omega^2)^2 + 4\zeta^2\omega^2\omega_n^2}$$

Magnitude:

$$|H(j\omega)| = \sqrt{\mathrm{Re}(H(j\omega))^2 + \mathrm{Im}(H(j\omega))^2} = \frac{\omega_n^2}{\sqrt{(\omega_n^2 - \omega^2)^2 + 4\zeta^2\omega^2\omega_n^2}}$$

Cut-off frequency is where magnitude at *half-power point*, which is $\frac{1}{\sqrt{2}}$ (approximately at $-3\,\mathrm{dB}$):

$$20\log_{10} K = 20\log_{10}\frac{1}{\sqrt{2}} \approx -3\,\mathrm{dB}$$

So the equation is:

$$|H(j\omega)| = \frac{1}{\sqrt{2}}$$

$$\frac{\omega_n^2}{\sqrt{(\omega_n^2 - \omega^2)^2 + 4\zeta^2\omega^2\omega_n^2}} = \frac{1}{\sqrt{2}}$$

Solve it for $\omega = \omega_c$:

$$\omega_c = \omega_n\sqrt{1 - 2\zeta^2 + \sqrt{4\zeta^4 - 4\zeta^2 + 2}}$$

All calculations were done in `Maxima`, see listing 3.

Listing 3: Calculate cut-off frequency

```
_H(s) := omega_n^2 / (s^2 + 2*zeta*omega_n*s + omega_n^2)$
H(omega) := subst([s = %i*omega], _H(s))$
R(omega) := realpart(H(omega))$
X(omega) := imagpart(H(omega))$
A(omega) := cabs(H(omega))$
/* Print values */
H(omega);R(omega);X(omega);A(omega);
eq1: A(omega) = 1/sqrt(2)$
sol: solve(eq1, omega)$
omega_c: rhs(sol[2]);
```

3 Simulating the filter

3.1 Simplification

To simplify calculations, let's make an assumption:

$$R = R_1 = R_2$$

$$C = C_1 = C_2$$

Then transfer function parameters will be:

$$\omega_n = \frac{1}{RC}$$

$$\zeta = 1.5$$

The quality factor:

$$Q = \frac{1}{3}$$

$Q < \frac{1}{2}$, so the system is overdamped (doesn't oscillate).

For calculated ζ and ω_N, we can recalculate out ω_c:

$$\omega_c = \frac{\omega_n}{2.672}$$

The cut-off frequency can be calculated:

$$f_c = \frac{\omega_c}{2\pi} = \frac{\omega_n}{16.789} = \frac{1}{16.789RC}$$

3.2 RC calculations

Let's calculate R and C values for given cut-off frequency:

$$f_c \geqslant 16\,\mathrm{kHz}$$

Let's choose resistors from E24 series, and capacitors from E6 series, so that $f_c \geqslant 16\,\mathrm{kHz}$:

$$R = 11\,\mathrm{k\Omega}$$

$$C = 330\,\mathrm{pF}$$

Then real cut-off frequency is:

$$f_c = \frac{1}{16.789RC} = 16.408\,\mathrm{kHz}$$

Or in angular form:

$$\omega_c = 2\pi f_c = 103\,094.5\,\frac{\mathrm{rad}}{\mathrm{s}}$$

3.3 Simulation in Maxima

Listing 4 shows the way to plot bode diagrams in `Maxima`.

Listing 4: Bode plot

```
R: 11e3$
C: 330e-12$
omega_n: 1/R/C$
zeta: 1.5$
H(s) := omega_n^2 / (s^2 + 2*zeta*omega_n*s + omega_n^2)$
load("bode")$
omega_from: 2*%pi*100$
omega_to: 2*%pi*100000$
bode_gain(H(s), [omega, omega_from, omega_to])$
```

The resulting plot is shown on figure 2.



Figure 2: Bode gain plot from Maxima

One can also use `COMA` package in `Maxima` for Control Engineering calculations and visualization.

3.4 Simulation in Qucs



Figure 3: Scheme from Qucs



Figure 4: Plot from Qucs